
Example nº 6
Gravity dam

CivilFEM Manual of Advanced Examples

Example nº 6 – Table of Contents

- 6 EXAMPLE Nº 6: GRAVITY DAM..... 1
 - 6.1 AIM 1
 - 6.2 DESCRIPTION OF THE EXAMPLE 1
 - 6.3 RESULTS TO BE OBTAINED 3
 - 6.4 CALCULATION LOG 4
 - 6.4.1 Introduction 4
 - 6.5 STRUCTURAL ANALYSIS 6
 - 6.5.1 Auxiliary macro: MACRO_PARAMETERS_VERTEX 6
 - 6.5.2 Auxiliary macro: MACRO_LINES 8
 - 6.5.3 Auxiliary macro: MACRO_AREAS 10
 - 6.5.4 Auxiliary macro: MACRO_SOLSEC 10
 - 6.5.5 Section A: Preprocess 14
 - 6.5.6 Section A: Analysis 20
 - 6.5.7 Section A: Postprocess 21
 - 6.6 SEEPAGE ANALYSIS 23
 - 6.6.1 Auxiliary macro: MACRO_PREP7_FOR_SEEP_MOD 23
 - 6.6.2 Section B.1: Preprocess 26
 - 6.6.3 Section B.1: Analysis 27
 - 6.6.4 Section B.1: Postprocess 28
 - 6.6.5 Section B.2: Preprocess 29
 - 6.6.6 Section B.2: Analysis 30
 - 6.6.7 Section B.2: Postprocess 30
 - 6.7 COMBINED MODEL 31
 - 6.7.1 Section C: Preprocess 31
 - 6.7.2 Section C: Analysis 31
 - 6.7.3 Section C: Postprocess 31
 - 6.8 RESULTS 32
 - 6.8.1 Maximum stresses in the terrain at points A and B at the surroundings of the underground gallery considering empty dam 32
 - 6.8.2 Stresses (Pa) and vertical displacements (m) considering empty dam 33

6.8.3	Maximum stresses in the terrain at points A and B at the surroundings of the underground gallery considering full water level	35
6.8.4	Stresses (Pa) and vertical displacements (m) considering full water level.....	36
6.8.5	Flow into the circular gallery and water gradient under the dam	38
6.8.6	Flow into the circular gallery and water gradient under the dam when an impermeable wall is built	39
6.9	SUMMARY	40

6 EXAMPLE N° 6: GRAVITY DAM

6.1 AIM

The aim of this example is to build a complex model that it groups characteristics of dam engineering, geotechnics, seepage analysis and design of reinforced concrete structures.


For it, a gravity dam that leans on a relatively complex foundation has been chosen. The terrain has two separated layers by a non horizontal plane, on which a reinforced concrete circular gallery has been excavated.

6.2 DESCRIPTION OF THE EXAMPLE

The following figure represents a concrete dam with a foundation formed by two layers clearly differentiated, with different permeability materials.

It is a rolled compacted concrete dam and the construction joints are not going to be grouted.

The gallery underneath the dam is built with porous concrete (drain pipe).



The properties of the concrete dam are:

$$E = 2.1 \cdot 10^6 \text{ T/m}^2$$

$$\nu = 0.22$$

$$\gamma = 2.35 \text{ T/m}^3$$

The upper material is a sand clay with the following properties:

$$E = 10000 \text{ T/m}^2$$

$$\nu = 0.35$$

$$\varphi = 30^\circ$$

$$c = 2 \text{ T/m}^3$$

$$\delta = 15^\circ$$

$$k = 10^{-5} \text{ m/s}$$

$$\gamma_d = 2.20 \text{ T/m}^3$$

and the lower material is a healthy granite of RMR = 80, with a very low permeability $k_h = k_v = 10^{-9} \text{ m/s}$ and a wave velocity of $a = 2500 \text{ m/s}$.

The geometric parameters are:

Heights		Lengths		Slopes	
z_1	1021.72 m	e_{top}	4.00 m	t_{up}	0.05
z_2	$z_1 - 0.90$ m	a	2.00 m	t_{down}	0.75
z_3	$z_1 - 70.00$ m	b	2.50 m		
z_4	$z_3 - 20.00$ m	D	4.00 m		
z_5	$z_1 - 3.00$ m	e_{toe}	30.00 m		
z_6	$z_3 - 60.00$ m	l_{horiz}	150.00 m		
z_{inf}	$z_6 - 80.00$ m	r_{horiz}	150.00 m		

The underground gallery has a wall thickness of $t_{hk} = 25$ cm; and D_h represents the inner diameter. The dimensions of the service gallery inside the dam are: $a_1 = 5.00$, $a_2 = 4.00$, $b_1 = 3.00$ and $b_2 = 2.50$, all of them in meters.

6.3 RESULTS TO BE OBTAINED

A. STRUCTURAL MODEL

A.1. Maximum stresses in the terrain at points A and B and at the surroundings of the underground gallery.

A.1.1. Considering empty dam.

A.1.2. Considering the water at its maximum height.

A.2. Obtain the necessary reinforcement around the gallery of the dam.

A.3. Design the reinforcement of the circular gallery, supposing that it is a circular ring with HA-35 concrete (Spanish code EHE).

B. SEEPAGE MODEL

B.1. Filtered flow into the circular gallery and hydraulic gradient at the upstream foot under the dam.

B.2. Obtain these same values, when a 40 m length impermeable wall is constructed from the upstream foot. The thickness of the wall is 2.5 m.

C. COMBINED MODEL. Obtain the vertical effective stresses in the foundations.

6.4 CALCULATION LOG

6.4.1 Introduction

At the beginning of the resolution of **section A**, the structural materials are defined, changing their properties to those defined in the example. It is not necessary to modify the permeability parameters of each material, because this section will perform a structural analysis.

The static elasticity modulus of the granite can be obtained by means of the correlations implemented in CivilFEM, using the wave velocity as well as the confinement stress created by concrete dam.

For the calculation a structural 2D plane strain element is used (PLANE42).

Because of the similarity between the different sections of this example, and to prevent repeating the same code several times, *APDL* macros with the parameters definition and creation of the keypoints and lines of the model. These lines will have the number division assigned so that the mesh is fitted properly. Another macros are created for the construction of the areas and its mesh, and for the definition of solid sections.

In order to extend the code checking capabilities to generic 2D/3D models, the solid sections are used. This entities are plane cross sections associated with finite elements models which do not use unidimensional (beam) elements, that have the same capacities of reinforcement definition, checking and design, etc. than the beam cross sections. It is necessary to select the plane of nodes that define the location of the section and the elements that will give its properties.

In order to capture a solid section it is necessary to establish a local cartesian coordinates system that defines the plane of nodes of the section. After capturing the section and in order to continue working with the model, it is important to return to reselect all the entities of the complete model (nodes, elements, etc.) and to activate the necessary coordinates system in each case (in this example the global cartesian is always used).

The creation of a solid section implies the automatic generation of a cross section associated to it. On this cross section the reinforcement that will be used for code checking are defined.

Before solving the created model the boundary conditions of contour and the loads of each load step must be set. In order to include self weight in the model, the initial stresses of the terrain are calculated so that it has an initial stress state, in equilibrium with its weight.

Once the model is solved, the stresses at the points specified in the example description, around the circular gallery, are read. For each one of the points two stress values are taken, one at the border terrain-concrete, and other moved slightly away from the gallery. Also the code checking of the previously captured solid sections is performed.

In **section B** only the permeable materials are defined (all the materials except those of the dam), since a seepage analysis will be performed.

In sections B.1 and B.2 the filtered flow to the circular gallery is requested. For that reason the concrete material used to defined the gallery is replaced by another geotechnical material with equivalent permeability.

The parameters that define the model geometry are set, without considering neither the dam or its gallery (its behavior will be entered through the boundary conditions). Nevertheless, it is necessary to consider at the time of creating the solid model, the geometry corresponding to the wall that is described in section B.2. Considering it from the beginning will allow use the same the macro for the lines generation for both sections.

For section B.1 the areas and mesh are created without considering the dam, from the defined keypoints and the assigned divisions of the created lines. The area of the wall of section B.2 is filled with the corresponding geotechnical materials (just as it is before being built).

For section B.2 the example requests the same results as in the previous section, but considering an impermeable wall of 40 m of depth from the dam. In this case everything done in the previous section is valid, except than now the areas of the wall must not be meshed, in order to simulate the impermeable behavior.

The boundary conditions based on the piezometric heights are set. At the upper stream surface, the border is in direct contact with the water and is the origin of the seepage. In this line $H = H_0$, where H_0 is the total water height. The dripping surface has atmospheric pressure and therefore $H = \text{height}$.

Next, for both sections B.1 and B.2, the analysis is performed using CivilFEM seepage elements.

Once the model is solved, the results file (with *.SEEP* extension) is read.

In order to go through **section C** some of the previously created log files will be used.

The effective stresses are calculated in the stress-strain model, subtracting the water pore pressures. CivilFEM allows to transfer the obtained water pressures from a seepage analysis to a structural analysis of with independency in the meshes of both models.

In this case the structural part is identical to the one of section A and the results of interstitial pressures have been already calculated in section B.1. CivilFEM will automatically read the pore pressures results file when the representation of the effective stresses is requested. The seepage results file must have the same *jobname* and must be located in the same working folder used to solve the model in this section.

6.5 STRUCTURAL ANALYSIS

6.5.1 Auxiliary macro: MACRO_PARAMETERS_VERTEX

```

! MACRO_PARAMETERS_VERTEX: Define parameters & vertex for example 6,
! structural analysis

/PREP7

! Parameters
etop = 4.00      ! width of dam's top
dist1 = 4.5      ! Distance between dam bottom left corner and bottom
                 ! left corner of dam gallery outer contour
a1 = 5.00       ! a1,b1 = dimensions of dam gallery. Outer contour
b1 = 3.00
a2 = 4.00       ! a2,b2 = dimensions of dam gallery. Inner contour
b2 = 2.5
D = 4.00        ! D = inner diameter of circular gallery
Thk = 0.25      ! Thk = thickness for circular gallery --> Outer
                 ! diameter of circular gallery = D+2*Thk

etoe = 30.00    ! horizontal distance from dam bottom right corner
                 ! to granite vertex

tup = 0.05
tdown = 0.75
lhoriz = 150
rhoriz = 150
! Heights
totalh = 1021.720 ! Total height of the model, from global system
                 ! origin (Y direction)
                 ! m.s.n.m
Z1 = totalh
Z2 = Z1 - 1.5
Z3 = Z1 - 70.00
Z4 = Z3 - 20.00
Z5 = Z1 - 3.00
Z6 = Z3 - 60.00
Zinf = Z6 - 80.00
baseleft = (Z1-Z3)*tup ! horizontal projection of left dam slope
wbaseleft = (Z2-Z3)*tup ! horizontal projection of left dam slope
                 ! with water
baseright = (Z5-Z3)*tdown ! horizontal projection of right dam slope
! Total width of the model, from global system origin (X direction)
totalw = lhoriz + baseleft + etop + baseright + etoe + rhoriz
dist2 = dist1 + (b1-b2) ! Distance between dam bottom left corner
                 ! and bottom left corner of dam gallery
                 ! inner contour

c = dist2-dist1
!-----
! Terrains vertex coordinates
X1 = 0
Y1 = Zinf
X2 = totalw
Y2 = Y1
X3 = 0
Y3 = Z6
X4 = 0
Y4 = Z3
X5 = lhoriz
Y5 = Y4
X6 = X5 + baseleft + etop + baseright
Y6 = Y4
X7 = X6 + etoe

```

```

Y7 = Y4
X8 = X2
Y8 = Y4
! Dam vertex coordinates
X9 = X5 + wbaseleft      ! X coordinate of water level
Y9 = Z2
X10 = X5 + baseleft + etop
Y10 = Z5
X11 = X5 + baseleft
Y11 = totalh
X12 = X10
Y12 = Y11
! Dam gallery vertex coordinates
X13 = X5 + dist1
Y13 = Y5 + dist1
X14 = X13 + c
Y14 = Y13
X15 = X13 + a1 - C
Y15 = Y13
X16 = X13 + a1
Y16 = Y13
X17 = X14
Y17 = Y5 + dist2
X18 = X17 + a2
Y18 = Y17
X19 = X13
Y19 = Y13 + b1
X20 = X14
Y20 = Y19
X21 = X15
Y21 = Y19
X22 = X16
Y22 = Y19
X23 = X14 + a2/2
Y23 = Y20 + a2/2
X24 = X23
Y24 = Y20 + a1/2
X38 = X23      ! Arc center coordinates K,38
Y38 = Y19
! Inner circular gallery coordinates
Y25 = Z4
X25 = X7*(1-((Y7-Y25)/(Y4-Y3)))      ! X coordinate of
                                      ! circular gallery center
X26 = X25 + D/2
Y26 = Y25
theta = ATAN((Y7-Y3)/X7)
X27 = X25 + (D/2)*COS(theta)
Y27 = Y25 + (D/2)*SIN(theta)
X28 = X25
Y28 = Y25 + D/2
X29 = X25 - D/2
Y29 = Y25
X30 = X25 - (D/2)*COS(theta)
Y30 = Y25 - (D/2)*SIN(theta)
X31 = X25
Y31 = Y25 - D/2
! Outer circular gallery coordinates
X32 = X25 + ((D/2)+Thk)
Y32 = Y25
X33 = X25 + ((D/2)+Thk)*COS(theta)
Y33 = Y25 + ((D/2)+Thk)*SIN(theta)
X34 = X25
Y34 = Y25 + ((D/2)+Thk)
X35 = X25 - ((D/2)+Thk)
Y35 = Y25
X36 = X25 - ((D/2)+Thk)*COS(theta)
Y36 = Y25 - ((D/2)+Thk)*SIN(theta)
X37 = X25

```

```

Y37 = Y25 - ((D/2)+Thk)
X39 = X25
Y39 = Y28+(Y34-Y28)/2
X40 = X35+(X29-X35)/2
Y40 = Y25
X41 = X25
Y41 = Y37+(Y31-Y37)/2
X42 = X26+(X32-X26)/2
Y42 = Y25
X43 = X15
Y43 = Y15+(Y18-Y15)/2
X44 = X21+(X22-X21)/2
Y44 = Y38
X45 = X38
Y45 = Y23+(Y24-Y23)/2
X46 = X19+(X20-X19)/2
Y46 = Y38
X47 = X14
Y47 = Y14+(Y17-Y14)/2

*DO,I,1,47
  K,I,X%I%,Y%I%
*ENDDO

```

6.5.2 Auxiliary macro: MACRO_LINES

```

! MACRO_LINES

! Lines & LESIZE
! Lines of granite area contour
LSEL,NONE
L,1,3
LESIZE,ALL,,10
LSEL,NONE
L,3,36
LESIZE,ALL,,37,0.1
LSEL,NONE
L,36,30
LESIZE,ALL,,2
! lines of inner circular gallery
*DO,II,26,30
LSEL,NONE
LARC,II,II+1,25,D/2
LESIZE,ALL,2*c
*ENDDO
LSEL,NONE
LARC,31,26,25,D/2
LESIZE,ALL,2*c
! Lines of outer circular gallery
*DO,II,32,36
LSEL,NONE
LARC,II,II+1,25,D/2+Thk
LESIZE,ALL,2*c
*ENDDO
LSEL,NONE
LARC,37,32,25,D/2+Thk
LESIZE,ALL,2*c
! Lines for capturing sections in circular gallery
LSEL,NONE
L,26,32
L,28,34
L,29,35
L,31,37
L,27,33
LESIZE,ALL,,2

```

```

LSEL,NONE
L,33,7
LESIZE,ALL,,,25,3.5 !0.1
LSEL,NONE
L,7,8
LESIZE,ALL,,,18
LSEL,NONE
L,8,2
LESIZE,ALL,,,15
LSEL,NONE
L,2,1
LESIZE,ALL,,,40
! Lines of clay area contour
LSEL,NONE
L,3,4
LESIZE,ALL,,,8
LSEL,NONE
L,4,5
LESIZE,ALL,,,20,0.1
LSEL,NONE
L,5,6
LESIZE,ALL,,,30,3
LSEL,NONE
L,6,7
LESIZE,ALL,,,6,1.5
! Lines of concrete area contour (dam)
LSEL,NONE
L,5,9
LESIZE,ALL,,,38,3.5
LSEL,NONE
L,9,11
LESIZE,ALL,,,1
LSEL,NONE
L,11,12
L,12,10
LESIZE,ALL,,,2
LSEL,NONE
L,10,6
LESIZE,ALL,,,32
! Lines of outer dam gallery contour
LSEL,NONE
L,13,14
L,15,16
LESIZE,ALL,c
LSEL,NONE
L,14,15
L,16,22
L,19,13
LARC,22,24,38,a1/2
LARC,24,19,38,a1/2
! Lines of inner dam gallery contour
L,17,18
L,18,21
L,20,17
LARC,21,23,38,a2/2
LARC,23,20,38,a2/2
LESIZE,ALL,2*c
! Lines for capturing sections in dam gallery
LSEL,NONE
L,24,23
L,19,20
L,14,17
L,15,18
L,21,22
L,13,17
L,16,18
LESIZE,ALL,,,2

```

6.5.3 Auxiliary macro: MACRO_AREAS

```

! MACRO_AREAS: Area definition & meshing
! Areas
! Granite area definition & meshing
ASEL,NONE
A,1,3,36,37,32,33,7,8,2
MAT,3
TYPE,1
REAL,10
AMESH,ALL
! Clay area
ASEL,NONE
A,3,4,5,6,7,33,34,35,36
MAT,2
TYPE,1
REAL,10
AMESH,ALL
! Circular gallery areas
ASEL,NONE
A,26,32,33,27
A,27,33,34,28
A,28,34,35,29
A,29,35,36,30
A,30,36,37,31
A,31,37,32,26
MAT,1
TYPE,1
REAL,10
AMESH,ALL
! Dam gallery areas
ASEL,NONE
A,5,9,11,12,10,6    ! Dam area
A,16,22,21,18
A,21,22,24,23
A,23,24,19,20
A,17,20,19,13
A,14,15,18,17
A,13,14,17
A,15,16,18
AOVLAP,ALL
ADELE,17,, ,1
MAT,4
TYPE,1
REAL,10
AMESH,ALL

```

6.5.4 Auxiliary macro: MACRO_SOLSEC

```

! MACRO_SOLSEC
! Circular Gallery Solid Sections
! Definition of the 1st solid section of circular gallery
KWPLAN,-1,39        ! Defines the working plane using three
                    ! keypoints
LSEL,S,LOC,X,X25,X25+0.0001 ! Select the line that contains
                           ! the section
LSEL,R,LOC,Y,Y25,1000
NSLL,S,1
NSEL,R,LOC,Y,Y28,Y39
ESLN,S
NSLE,S

```

```

NSEL,R,LOC,X,-1000,X25
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1,      ! Defines a local coordinate system at the
                       ! origin of the working plane
NSEL,R,LOC,X,0        ! Reselect only the nodes of the selected
                       ! elements that are in the side of section
                       ! to capture
~SLDSEC,1,1,11,Solid Section 1

ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 2nd solid section of circular gallery
KWPLAN,-1,42         ! Defines the working plane using three keypoints
WPRO,-90.0,0,0
LSEL,S,LOC,Y,Y25,Y25+0.0001 ! Select the line that contains the section
LSEL,R,LOC,X,X25,1000
NSLL,S,1
NSEL,R,LOC,Y,Y26,Y32
ESLN,S
NSLE,S
NSEL,R,LOC,Y,Y25,1000
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1,      ! Defines a local coordinate system at the
                       ! origin of the working plane

NSEL,R,LOC,X,0        ! Reselect only the nodes of the selected elements
                       ! that are in the side of section to capture
~SLDSEC,2,1,11,Solid Section 2      ! Solid section definition

ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 3rd solid section of circular gallery
KWPLAN,-1,41         ! Defines the working plane using three keypoints
WPRO,180.0,0,0
LSEL,S,LOC,X,X25,X25+0.0001      ! Select the line that contains the section
LSEL,R,LOC,Y,Y25,-1000
NSLL,S,1
NSEL,R,LOC,Y,Y41,Y31
ESLN,S
NSLE,S
NSEL,R,LOC,X,X25,1000
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1,      ! Defines a local coordinate system at the origin
                       ! of the working plane
NSEL,R,LOC,X,0        ! Reselect only the nodes of the selected elements
                       ! that are in the side of section to capture
~SLDSEC,3,1,11,Solid Section 3
/SHOW,CLOSE
ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 4th solid section of circular gallery
KWPLAN,-1,40         ! Defines the working plane using three keypoints
WPRO,90.0,0,0
LSEL,S,LOC,Y,Y25,Y25+0.0001      ! Select the line that contains the section
LSEL,R,LOC,X,-1000,X25
NSLL,S,1

```

```

NSEL,R,LOC,Y,Y40,Y29
ESLN,S
NSLE,S
NSEL,R,LOC,Y,Y25,-1000
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1,      ! Defines a local coordinate system at the
                      ! origin of the working plane
NSEL,R,LOC,X,0        ! Reselect only the nodes of the selected elements
                      ! that are in the side of section to capture
~SLDSEC,4,1,11,Solid Section 4      ! Solid section definition

ALLSEL,ALL
/REPLOT
CSYS,0

! Dam Gallery Solid Sections
! Definition of the 1st solid section of dam gallery
KWPLAN,-1,43          ! Defines the working plane using three keypoints
LSEL,S,LOC,X,X15,X15+0.0001 ! Select the line that contains the section
LSEL,R,LOC,Y,-1000,Y18
NSLL,S,1
NSEL,R,LOC,Y,Y43,Y18
ESLN,S
NSLE,S
NSEL,R,LOC,X,-1000,X15
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1,      ! Defines a local coordinate system at the
                      ! origin of the working plane
NSEL,R,LOC,X,0        ! Reselect only the nodes of the selected elements
                      ! that are in the side of section to capture
~SLDSEC,5,1,11,Solid Section 5

ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 2nd solid section of dam gallery
KWPLAN,-1,44          ! Defines the working plane using three keypoints
WPRO,90.0,0,0
LSEL,S,LOC,Y,Y38,Y38+0.0001 ! Seleccionamos la línea que contiene a
                          ! la sección
LSEL,R,LOC,X,X38,1000
NSLL,S,1
NSEL,R,LOC,X,X21,X44
ESLN,S
NSLE,S
NSEL,R,LOC,Y,-1000,Y38
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1,      ! Defines a local coordinate system at the
                      ! origin of the working plane
NSEL,R,LOC,X,0        ! Reselect only the nodes of the selected
                      ! elements that are in the side of section to capture
~SLDSEC,6,1,11,Solid Section 6

ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 3rd solid section of dam gallery
KWPLAN,-1,45          ! Defines the working plane using three keypoints
WPRO,180.0,0,0

```

```

LSEL,S,LOC,X,X38,X38+0.0001 ! Select the line that contains the section
LSEL,R,LOC,Y,Y38,1000
NSLL,S,1
NSEL,R,LOC,Y,Y23,Y45
ESLN,S
NSLE,S
NSEL,R,LOC,X,X38,1000
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1, ! Defines a local coordinate system at the origin
! of the working plane
NSEL,R,LOC,X,0 ! Reselect only the nodes of the selected elements
! that are in the side of section to capturer
~SLDSEC,7,1,11,Solid Section 7

ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 4th solid section of dam gallery
KWPLAN,-1,46 ! Defines the working plane using three keypoints
WPRO,-90.0,0,0
LSEL,S,LOC,Y,Y38,Y38+0.0001 ! Select the line that contains the section
LSEL,R,LOC,X,-1000,X38
NSLL,S,1
NSEL,R,LOC,X,X46,X20
ESLN,S
NSLE,S
NSEL,R,LOC,Y,Y38,1000
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1, ! Defines a local coordinate system at the origin
! of the working plane
NSEL,R,LOC,X,0 ! Reselect only the nodes of the selected elements
! that are in the side of section to capture
~SLDSEC,8,1,11,Solid Section 8

ALLSEL,ALL
/REPLOT
CSYS,0

! Definition of the 5th solid section of dam gallery
KWPLAN,-1,47 ! Defines the working plane using three keypoints
WPRO,180.0,0,0
LSEL,S,LOC,X,X14,X14+0.0001 ! Select the line that contains the section
LSEL,R,LOC,Y,-1000,Y17
NSLL,S,1
NSEL,R,LOC,Y,Y47,Y17
ESLN,S
NSLE,S
NSEL,R,LOC,X,X14,1000
ESLN,S,1
EPLOT

CSWPLA,11,0,1,1, ! Defines a local coordinate system at the origin
! of the working plane
NSEL,R,LOC,X,0 ! Reselect only the nodes of the selected elements
! that are in the side of section to capture
~SLDSEC,9,1,11,Solid Section 9

ALLSEL,ALL
/REPLOT
CSYS,0

```

6.5.5 Section A: Preprocess

```

FINISH
~CFCLEAR,,1

~CFACTIV,GETC,Y      ! Activate geotechnical module
~SEEPAGE,NONE

/TITLE,Dam analysis

~CODESEL,EC3-92,EHE,EC2-91,,EC8-94

! CivilFEM Setup
~UNITS,SI
! =====
/PREP7
! CivilFEM Preprocessor
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-35      ! Circular Gallery

~CFMP,4,LIB,CONCRETE,EHE,HA-20      ! Dam
~CFMP,4,USER
~CFMP,4,Concr ,ExLn,,2.0601e10      ! Elasticity modulus for linear
! analysis = 2.1E6 T/m^2 = 2.0601e10 N/m^2
~CFMP,4,DatGen,NUxy,,0.22           ! Poisson's ratio = 0.220
~CFMP,4,DatGen,GAM ,,,23035.5       ! Specific weight = 2.35 T/m^3 = 23035.5 N/m^3

~CFMP,2,LIB,SOIL,,SC
~CFMP,2,SOIL,KPLA,,0                ! Type of behavior: Elastic
~CFMP,2,SOIL,NUXYST,,0.35          ! Static Poisson modulus

~CFMP,2,SOIL,EXST,,9.81e7           ! Static elasticity modulus =
! 10.000 T/m^2 = 9.81e+007 N/m^2
~CFMP,2,SOIL,DELEFF,,15            ! Dilation = 15°
~CFMP,2,SOIL,GAMD,,21582           ! Dry specific weight = 2.20 T/m^3 = 21582 N/m^3

~CFMP,3,LIB,ROCK,,Granite,
~CFMP,3,ROCK,GSI,,80                ! RMR = 80
~CRLTUSE,3,3,1,ROCK,ExD,48,CivilFEM ! Applies a correlation to the rock dynamic
! elasticity modulus
~CRLTUSE,3,3,1,ROCK,ExSt,40,CivilFEM ! Applies a correlation to the rock Static
! elasticity modulus (sigma0<15Mpa (Heerden))


~CFMP,5,LIB,REINF,EHE,B 500 S       ! Reinforcement for both galleries
!-----
! Model definition
! Define parameters & vertex for problem 6, structural analysis
MACRO_PARAMETERS_VERTEX
!-----
! KP'S
! External contour KP's-->K1-K12
! Dam gallery KP's-->K13-K38
! Inner Circular gallery KP's-->K25-K31
! Outer circular gallery KP's-->K33-K47
*DO,II,1,47
  K,II,X%II%,Y%II%
*ENDDO

/PNUM,KP,1
/REPLOTT
!-----
! Lines & LESIZE
MACRO_LINES


ALLSEL,ALL
/PNUM,LINE,1
/PNUM,KP,1

```


/REPLOT
L PLOT



```
!-----
ET,1,PLANE42 ! used for 2-D modeling of solid structures. Plane strain
KEYOPT,1,3,2
R,10
!-----
! Areas definition & meshing
MACRO_AREAS
```



```
ALLSEL,ALL  
EPLOT  
!-----  
! Solid sections definition  
MACRO_SOLSEC
```



```

! Rectangular sections reinforcement
*DO, II, 1, 4
  ~SECMDF, II, FACES, PT_ADD, LOC, , 1, 0.125, -0.5
  ~SECMDF, II, FACES, PT_ADD, LOC, , 1, 0.125, 0.5
  ~SECMDF, II, FACES, PT_ADD, LOC, , 2, -0.125, 0.5
  ~SECMDF, II, FACES, PT_ADD, LOC, , 2, -0.125, -0.5
  ~RNFDEF, II, 1, 5, 1, 0, 20.000E-03, , , 16.00, 5, , , 4
  ~RNFDEF, II, 2, 5, 2, 0, 20.000E-03, , , 16.00, 5, , , 4
*ENDDO
*DO, II, 5, 9
  ~SECMDF, II, FACES, PT_ADD, LOC, , 1, 0.25, -0.5
  ~SECMDF, II, FACES, PT_ADD, LOC, , 1, 0.25, 0.5
  ~SECMDF, II, FACES, PT_ADD, LOC, , 2, -0.25, 0.5
  ~SECMDF, II, FACES, PT_ADD, LOC, , 2, -0.25, -0.5
  ~RNFDEF, II, 1, 5, 1, 0, 50.000E-03, , , 16.00, 5, , , 4
  ~RNFDEF, II, 2, 5, 2, 0, 50.000E-03, , , 16.00, 5, , , 4
*ENDDO
FINISH
    
```

- Preferences
- Preprocessor
- Solution
- General Postproc
- TimeHist Postpro
- Topological Opt
- ROM Tool
- Design Opt
- Prob Design
- Radiation Opt
- Run-Time Stats
- Session Editor
- Finish
- CivilFEM
- Civil Setup
 - Civil Preprocessor
 - Element Types
 - Materials
 - Cross Sections**
 - Shell Vertex
 - Member Properties
 - Beam & Shell Properties
 - Solid Section
 - Verif. Database
 - Update Database
 - Active Time
 - Load State
 - Geotech. mod.
 - Civil Solution
 - Civil Postprocessor
 - Seismic Design
 - Flac3D
 - Geotechnical module
 - Bridge and Civil Non Linear
 - Prestressed Concrete

Cross Sections Explorer

Cross Section list

Num	Name	Reference	Type	Shape
...	1	Reinfor Concrete	Reinforced concrete	G Generic
...	2	Reinfor Concrete	Reinforced concrete	G Generic
...	3	Reinfor Concrete	Reinforced concrete	G Generic
...	4	Reinfor Concrete	Reinforced concrete	G Generic
...	5	Reinfor Concrete	Reinforced concrete	G Generic
...	6	Reinfor Concrete	Reinforced concrete	G Generic
...	7	Reinfor Concrete	Reinforced concrete	G Generic
...	8	Reinfor Concrete	Reinforced concrete	G Generic
...	9	Reinfor Concrete	Reinforced concrete	G Generic


Active time: ActTime 28.00 Days

Buttons: Hot Rolled, Steel by Dimensions, Steel by Plates, Concrete, User Data Base, Modify, Copy, Delete, List, Save, Merge, Capture 2D Mesh, Export 2D Mesh, Help, Exit

Section 5:

Section Select Edit List View


Number 5 Name




Properties:


- A 500.000E-03 m²
- Ixx 52.083E-03 m⁴
- Iyy 41.667E-03 m⁴
- Izz 10.417E-03 m⁴
- Wy 83.333E-03 m³
- Wz 41.667E-03 m³

Contour View



- Preferences
- Preprocessor
- Solution
- General Postproc
- TimeHist Postpro
- Topological Opt
- ROM Tool
- Design Opt
- Prob Design
- Radiation Opt
- Run-Time Stats
- Session Editor
- Finish
- CivilFEM
- Civil Setup
- Civil Preprocessor
 - Element Types
 - Materials
 - Cross Sections
 - Shell Vertex
 - Member Properties
 - Beam & Shell Properties
 - Solid Section**
 - Verif. Database
 - Update Database
 - Active Time
 - Load State
 - Geotech. mod.
- Civil Solution
- Civil Postprocessor
- Seismic Design
- Flac3D
- Geotechnical module
- Bridge and Civil Non Linear
- Prestressed Concrete





6.5.6 Section A: Analysis


```

/SOLU

! Boundary conditions
DL,1,,UX,0
DL,25,,UX,0
DL,24,,UY,0
DL,23,,UX,0
DTRAN      ! Transfers solid model DOF constraints to the FEM
EPLLOT
!-----
! A. STRUCTURAL ANALYSIS
! A.1.1. EMPTY DAM
ACEL,,9.81           ! Self weight
ESEL,S,MAT,,2
ESEL,A,MAT,,3
EPLLOT
~TIS,IniStr,,-Y      ! Defines terrain initial stress
ISFILE,,IniStr, IST,
ALLSEL,ALL
SOLVE
!-----
! A.1.2. FILLED DAM
gammaw = 9810 ! N/m^3
SFL,26,PRES,gammaw*(Y9-Y5), ! Vertical load over the soil
SFGRAD,PRES,0,Y,Y5,-gammaw ! Pressure over the dam
SFL,29,PRES,gammaw*(Y9-Y5),
SFTRANS           ! Transfer the solid model surface
                  ! loads to the finite element model
                  ! on displays
/PSSF,PRES,NORM,2,0,1 ! Shows surface load symbols on model
                  ! displays
/PBF,DEFA, ,1     ! Shows magnitude of body force loads
                  ! on displays
/PIC,DEFA, ,1    ! Shows initial conditions on elements
                  ! as contours on displays
/PSYMB,DOT,1     ! Larger symbols displayed for node
                  ! and keypoint locations EPLLOT
    
```

SOLVE

FINISH



6.5.7 Section A: Postprocess

```

/POST1
! CivilFEM Postprocessor
!
-----
~CFSET,,1,1          ! Read first load step
/GRAPHICS,FULL       ! Display all model geometry and results
/DSCALE,ALL,1.0
PLNSOL, S,Y, 0,1.0
PLNSOL, U,Y, 0,1.0

! A1 point Stress. Clay-Circular_Gallery
! B1 point Stress. Granite-Circular_Gallery
*GET,StressA1,NODE,NODE(X34,Y34,0),S,Y
*GET,StressB1,NODE,NODE(X37,Y37,0),S,Y

! A2 point Stress. Clay
! B2 point Stress. Granite
*GET,StressA2,NODE,NODE(X34,Y34+8*Thk,0),S,Y
*GET,StressB2,NODE,NODE(X37,Y37-4*Thk,0),S,Y

! Bending + Axial checking for rectangular sections according to EHE
~CHKCON,2DB,,0      ! Checks axial force + bending cross sections
~CFSET,1           ! Read first alternative
~PRCON,DETAILED,ALLS, , ! Lists results of concrete checking and
                        ! design for all solid sections
    
```

```

!-----
~CFSET,,2,1          ! Read second load step
/GRAPHICS,FULL       ! Display all model geometry and results
/DSCALE,ALL,1.0
PLNSOL, S,Y, 0,1.0
PLNSOL, U,Y, 0,1.0

! A1 point Stress. Clay-Circular_Gallery
! B1 point Stress. Granite-Circular_Gallery
*GET,StressA1,NODE,NODE(X34,Y34,0),S,Y
*GET,StressB1,NODE,NODE(X37,Y37,0),S,Y

! A2 point Stress. Clay
! B2 point Stress. Granite
*GET,StressA2,NODE,NODE(X34,Y34+8*Thk,0),S,Y
*GET,StressB2,NODE,NODE(X37,Y37-4*Thk,0),S,Y

! Bending + Axial checking for rectangular sections according to EHE
~CHKCON,2DB,,0      ! Checks axial force + bending cross sections
~CFSET,1            ! Read first alternative
~PRCON,DETAILED,ALLS, ! Lists results of concrete checking and
                    ! design for all solid sections

```

6.6 SEEPAGE ANALYSIS

6.6.1 Auxiliary macro: MACRO_PREP7_FOR_SEEP_MOD

```

! MACRO_PREP7_FOR_SEEP_MOD: macro that includes: Materials, parameters,
! model (KP's & lines), Element type definition.
! Used in problem 6 seep analysis sections

! Materials
~CFMP,1,LIB,SOIL,,SC
~CFMP,1,SOIL,KY,,1e-5      ! Permeability Y

~CFMP,2,LIB,ROCK,,Granite,
~CFMP,2,ROCK,KX,,1e-9      ! Permeability X
~CFMP,2,ROCK,KY,,1e-9      ! Permeability Y

~CFMP,3,LIB,ROCK,,Kaolin,  ! Material for circular gallery -
                           ! Permeability equivalent to concrete
~CFMP,3,ROCK,KX,,1e-10     ! Permeability X
~CFMP,3,ROCK,KY,,1e-10     ! Permeability Y

!-----
! Model definition
! Parameters
etop   = 4.00      ! width of dam's top
D      = 4.00      ! D = inner diameter of circular gallery
Thk    = 0.25      ! Thk = thickness for circular gallery -->
                           ! Outer diameter of circular gallery = D+2*Thk
wthk   = 2.5       ! Wall thickness
whgt   = 40.00     ! Wall height
etoe   = 30.00     ! horizontal distance from dam bottom right
                           ! corner to granite vertex

tup    = 0.05
tdown  = 0.75
pbase  = 0.01
lhoriz = 150
rhoriz = 150
! Heights
totalh = 1021.720  ! Total hight of the model, from global system
                           ! origin (Y direction)
Z1     = totalh    ! m.s.n.m
Z2     = Z1 - 1.5
Z3     = Z1 - 70.00
Z4     = Z3 - 20.00
Z5     = Z1 - 3.00
Z6     = Z3 - 60.00
Zinf   = Z6 - 80.00
baseleft = (Z1-Z3)*tup  ! horizontal projection of left dam slope
wbaseleft = (Z2-Z3)*tup  ! horizontal projection of left dam slope
                           ! with water
baseright = (Z5-Z3)*tdown ! horizontal projection of right dam slope
! Total width of the model, from global system origin (X direction)
totalw = lhoriz + baseleft + etop + baseright + etoe + rhoriz
c      = 0.5         ! element size in circular gallery
!-----
! Terrains vertex coordinates
X1 = 0
Y1 = Zinf
X2 = totalw
Y2 = Y1
X3 = 0
Y3 = Z6

```

```

X4 = 0
Y4 = Z3
X5 = lhoriz
Y5 = Y4
X6 = X5 + wthk
Y6 = Y4
X7 = totalw - rhoriz
Y7 = Y4
X8 = X2
Y8 = Y4
X9 = X5
Y9 = Y5 - ((Y4 - Y3) * ((X7 - X5) / X7))
X10 = X6
Y10 = Y6 - ((Y4 - Y3) * ((X7 - X6) / X7))
X11 = X5
Y11 = Y4 - whgt
X12 = X6
Y12 = Y11
X30 = X5 + baseleft + etop + baseright
Y30 = Y4
! Inner circular gallery coordinates
Y13 = Z4
X13 = X7 * (1 - ((Y7 - Y13) / (Y4 - Y3)))    ! X coordinate of circular
                                           ! gallery center
X14 = X13 + D/2
Y14 = Y13
theta = ATAN((Y7 - Y3) / X7)
X15 = X13 + (D/2) * COS(theta)
Y15 = Y13 + (D/2) * SIN(theta)
X16 = X13
Y16 = Y13 + D/2
X17 = X13 - D/2
Y17 = Y13
X18 = X13 - (D/2) * COS(theta)
Y18 = Y13 - (D/2) * SIN(theta)
X19 = X13
Y19 = Y13 - D/2
! Outer circular gallery coordinates
X20 = X13 + ((D/2) + Thk)
Y20 = Y13
X21 = X13 + ((D/2) + Thk) * COS(theta)
Y21 = Y13 + ((D/2) + Thk) * SIN(theta)
X22 = X13
Y22 = Y13 + ((D/2) + Thk)
X23 = X13 - ((D/2) + Thk)
Y23 = Y13
X24 = X13 - ((D/2) + Thk) * COS(theta)
Y24 = Y13 - ((D/2) + Thk) * SIN(theta)
X25 = X13
Y25 = Y13 - ((D/2) + Thk)
X26 = X13
Y26 = Y16 + (Y22 - Y16) / 2
X27 = X23 + (X17 - X23) / 2
Y27 = Y13
X28 = X13
Y28 = Y25 + (Y19 - Y25) / 2
X29 = X14 + (X20 - X14) / 2
Y29 = Y13
!-----
! KP'S
*DO, II, 1, 30
  K, II, X%II%, Y%II%
*ENDDO

/PNUM, KP, 1
KPLOT
!-----
! Lines & LESIZE

```

```

! Lines of granite area contour
LSEL,NONE
L,1,3
LESIZE,ALL,,8
LSEL,NONE
L,3,9
LESIZE,ALL,,28,0.2
LSEL,NONE
L,9,11
L,12,10
LESIZE,ALL,,7
LSEL,NONE
L,5,6
L,9,10
L,11,12
LESIZE,ALL,,2
LSEL,NONE
L,10,24
LESIZE,ALL,,3
! lines of inner circular gallery
*DO,II,14,18
LSEL,NONE
LARC,II,II+1,13,D/2
LESIZE,ALL,3*c
*ENDDO
LSEL,NONE
LARC,19,14,13,D/2
LESIZE,ALL,3*c
! Lines of outer circular gallery
*DO,II,20,24
LSEL,NONE
LARC,II,II+1,13,D/2+Thk
LESIZE,ALL,3*c
*ENDDO
LSEL,NONE
LARC,25,20,13,D/2+Thk
LESIZE,ALL,3*c
LSEL,NONE
L,14,20
L,15,21
L,16,22
L,17,23
L,18,24
L,19,25
LESIZE,ALL,,2

LSEL,NONE
L,21,7
LESIZE,ALL,,36
LSEL,NONE
L,7,8
LESIZE,ALL,,30,2
LSEL,NONE
L,8,2
LESIZE,ALL,,15
LSEL,NONE
L,2,1
LESIZE,ALL,,45, !-0.2
! Lines of clay area contour
LSEL,NONE
L,3,4
LESIZE,ALL,,9
LSEL,NONE
L,4,5
LESIZE,ALL,,28,0.2
LSEL,NONE
L,5,9
L,6,10

```

```

LESIZE,ALL,,9
LSEL,NONE
L,6,30
LESIZE,ALL,,20
LSEL,NONE
L,30,7
LESIZE,ALL,,16

ALLSEL,ALL
/PNUM,LINE,1
/PNUM,KP,1
/REPLOT
LPLOT
R,10

```

6.6.2 Section B.1: Preprocess

```

FINISH
~CFCLEAR,,1


~CFACTIV,GETC,Y ; Activate geotechnical module
~SEEPAGE,2D ! seepage element type PLANE42-SEEP is defined

/TITLE,Dam analysis

~CODESEL,EC3-92,EHE,EC2-91,,EC8-94

! CivilFEM Setup
~UNITS,SI
! =====
/PREP7
! CivilFEM Preprocessor
! macro that includes: Materials, parameters, model (KP's & lines),
! Element type definition
MACRO_PREP7_FOR_SEEP_MOD

```



```

! Areas
! granite retaining wall area definition & meshing
ASEL,NONE
A,9,11,12,10
MAT,2
TYPE,1
REAL,10
AMESH,ALL
! Granite area definition & meshing
ASEL,NONE
A,1,3,9,11,12,10,24,25,20,21,7,8,2
MAT,2
TYPE,1
REAL,10
AMESH,ALL
! Clay retaining wall area definition & meshing
ASEL,NONE
A,9,10,6,5
MAT,1
TYPE,1
REAL,10
AMESH,ALL
! Clay area. Soil Meshing (left part)
ASEL,NONE
A,3,4,5,9
MAT,1
TYPE,1
REAL,10
AMESH,ALL
! Clay area. Soil Meshing (right part)
ASEL,NONE
A,10,6,30,7,21,22,23,24
MAT,1
TYPE,1
REAL,10
AMESH,ALL
! Circular gallery areas
ASEL,NONE
A,14,20,21,15
A,15,21,22,16
A,16,22,23,17
A,17,23,24,18
A,18,24,25,19
A,19,25,20,14
MAT,3
TYPE,1
REAL,10
AMESH,ALL

ALLSEL,ALL
EPLOT

FINISH

```

6.6.3 Section B.1: Analysis


```

/SOLU

! Boundary conditions
~DLHEAD,ADD,32,Z2 ! Boundary with a constant head.
~DLHEAD,ADD,36,Y4
~DLHEAD,ADD,28,Y4
*DO,I,9,14
  ~DLHEAD,ADD,I,Y4
*ENDDO

```

```
ALLSEL,ALL
DTRAN      ! Transfers solid model DOF constraints to the FEM
/PBC,ALL, ,1
EPLOT
```



```
! Solver
~LPSOLVE  ! Starts the finite elements model solution utilizing
           ! CivilFEM's seepage elements

FINISH
```

6.6.4 Section B.1: Postprocess

```
/POST1
FILE,FILE,SEEP
SET,1

/GRAPHICS,FULL      ! Display all model geometry and results
/DSCALE,1,OFF
~ISOBAR             ! Plot pore water pressure results
~LPRNSOL,GRAD,Y    ! List the nodal results when the CivilFEM's
                   ! seepage commands have been used

~PLSEEP,Q,VECTOR   ! Vectorial flow

! Transfers the pore water pressure results to the slope
! calculation routines
~WTSLP             ! This command automatically generates a file
                   ! called jobname.PRESS that will be used in Example6.C.
```

6.6.5 Section B.2: Preprocess

```

FINISH
~CFCLEAR,,1

~CFACTIV,GETC,Y ; Activate geotechnical module
~SEEPAGE,2D      ! Seepage element type PLANE42-SEEP is defined

/TITLE,Dam analysis

~CODESEL,EC3-92,EHE,EC2-91,,EC8-94

! CivilFEM Setup
~UNITS,SI
! =====
/PREP7
! CivilFEM Preprocessor
! macro that includes: Materials, parameters, model (KP's & lines),
! Element type definition
MACRO_PREP7_FOR_SEEP_MOD

! Areas
! Granite area definition & meshing
ASEL,NONE
A,1,3,9,11,12,10,24,25,20,21,7,8,2
MAT,2
TYPE,1
REAL,10
AMESH,ALL
! Clay area. Soil Meshing (left part)
ASEL,NONE
A,3,4,5,9
MAT,1
TYPE,1
REAL,10
AMESH,ALL
! Clay area. Soil Meshing (right part)
ASEL,NONE
A,10,6,30,7,21,22,23,24
MAT,1
TYPE,1
REAL,10
AMESH,ALL
! Circular gallery areas
ASEL,NONE
A,14,20,21,15
A,15,21,22,16
A,16,22,23,17
A,17,23,24,18
A,18,24,25,19
A,19,25,20,14
MAT,3
TYPE,1
REAL,10
AMESH,ALL


ALLSEL,ALL
EPLOT

FINISH

```

6.6.6 Section B.2: Analysis

The log file is the same as the one used in the chapter 6.6.3



6.6.7 Section B.2: Postprocess

```

/POST1
FILE, FILE, SEEP
SET, 1

/GRAPHICS, FULL      ! Display all model geometry and results
/DSCALE, 1, OFF
~ISOBAR              ! Plot pore water pressure results
~LPRNSOL, GRAD, Y    ! List the nodal results when the CivilFEM's
                     ! seepage commands have been used

~PLSEEP, Q, VECTOR   ! Vectorial flow

```

6.7 COMBINED MODEL

6.7.1 Section C: Preprocess

The log file is the same as the one used in the chapter 6.5.5

6.7.2 Section C: Analysis

The log file is the same as the one used in the chapter 6.5.6

6.7.3 Section C: Postprocess

```

/POST1
/GRAPHICS,FULL      ! Display all model geometry and results
ESEL,U,MAT,,1,4,3  ! Unselect Dam and circular gallery elements

~CFSET,,1,1        ! Read first load step
/DSCALE,ALL,1.0    ! True scale
PLNSOL, S,Y, 0,1.0
~EFFPRES,Y

~CFSET,,2,1        ! Read second load step
/DSCALE,ALL,1.0    ! True scale
PLNSOL, S,Y, 0,1.0
~EFFPRES,Y

```

6.8 RESULTS

In the following pages a series of figures are shown with the obtained.


6.8.1 Maximum stresses in the terrain at points A and B at the surroundings of the underground gallery considering empty dam.

A1 point Stress. Clay-Circular_Gallery


B1 point Stress. Granite-Circular_Gallery



A2 point Stress. Clay

B2 point Stress. Granite



6.8.2 Stresses (Pa) and vertical displacements (m) considering empty dam.






6.8.3 Maximum stresses in the terrain at points A and B at the surroundings of the underground gallery considering full water level

A1 point Stress. Clay-Circular_Gallery



B1 point Stress. Granite-Circular_Gallery



A2 point Stress. Clay

B2 point Stress. Granite





6.8.4 Stresses (Pa) and vertical displacements (m) considering full water level







6.8.5 Flow into the circular gallery and water gradient under the dam



6.8.6 Flow into the circular gallery and water gradient under the dam when an impermeable wall is built



6.9 SUMMARY

With the accomplishment of this exercise a very ample range of engineering subjects have been studied:

- Structural characteristics:
 - Structural analysis.
 - Reinforced concrete.
- Seepage characteristics:
 - Seepage analysis.
 - Obtaining filtered flows.
 - Obtaining pore water pressures.

As well as the combination of both analysis to obtain effective stresses in the foundations.

The reader has been able, therefore, to verify the possibilities that CivilFEM offers in this kind of studies, as well as to learn and practice the way it should be carried out.